Connection one-form

$\omega^k{}_\lambda = \Gamma^k{}_{\lambda\mu}\,\Theta^\mu$ Γ: Christoffel symbol or connection coefficient

Torsion two-form

$\Sigma^\mu = d\Theta^\mu + \omega^\mu{}_k \wedge \Theta^k$

The mainstream of GR is torsion-free!

Algorithm...

- Find Metric Tensor !
- Use Following Relation:

$d g_{\mu\nu} = \omega^k{}_\mu g_{k\nu} + \omega^k{}_\nu g_{k\mu} = \omega_{\mu\nu} + \omega_{\nu\mu}$
with above relation and connection one-form, find connection coefficient as possible

- Use Zero-Torsion Condition
- Now you can find all of connection coefficients

However, with computer, it's okay to use pesky definition;
$\Gamma^\lambda{}_{\mu\nu} = \frac{1}{2} g^{\lambda\sigma}(\partial_\mu g_{\sigma\nu} + \partial_\nu g_{\sigma\mu} - \partial_\sigma g_{\mu\nu})$,

Ex. $\mathbb{E}^3$, polar coordinate

```
In[1]:= dim = 3;
        coord = {r, θ, ϕ};
        metric = {{1, 0, 0}, {0, r^2, 0}, {0, 0, r^2 Sin[θ]^2}}
Out[3]= {{1, 0, 0}, {0, r^2, 0}, {0, 0, r^2 Sin[θ]^2}}

        metric // MatrixForm
```

In[4]:= $\begin{pmatrix} 1 & 0 & 0 \\ 0 & r^2 & 0 \\ 0 & 0 & r^2\,\mathrm{Sin}[\theta]^2 \end{pmatrix}$

```
Out[4]= {{1, 0, 0}, {0, r^2, 0}, {0, 0, r^2 Sin[θ]^2}}

In[5]:= inversemetric = Simplify[Inverse[metric]]
```

Out[5]= $\left\{\{1, 0, 0\}, \left\{0, \frac{1}{r^2}, 0\right\}, \left\{0, 0, \frac{\mathrm{Csc}[\theta]^2}{r^2}\right\}\right\}$

recall that the definition of connection coefficient;
$\Gamma^\lambda{}_{\mu\nu} = \frac{1}{2} g^{\lambda\sigma}(\partial_\mu g_{\sigma\nu} + \partial_\nu g_{\sigma\mu} - \partial_\sigma g_{\mu\nu})$

```
In[6]:= affine := affine = Simplify[Table[(1/2) * Sum[(inversemetric[[i, s]]) *
            (D[metric[[s, j]], coord[[k]] ] +
              D[metric[[s, k]], coord[[j]] ] - D[metric[[j, k]], coord[[s]] ]),
           {s, 1, dim}],
         {i, 1, dim}, {j, 1, dim}, {k, 1, dim}] ]

In[43]:= affine[[3, 1, 3]]
```

Out[43]= $\frac{1}{r}$

Display it: Γ[1, 2, 3] stands for $\Gamma^1{}_{23}$

In[10]:=
```
listaffine :=
  Table[{ToString[Γ[i, j, k]], affine[[i, j, k]]},
   {i, 1, dim}, {j, 1, dim}, {k, 1, dim}]

TableForm[listaffine, TableSpacing → {3, 3}]
```

Γ[1, 1, 1] 0	Γ[1, 2, 1] 0	Γ[1, 3, 1] 0
Γ[1, 1, 2] 0	Γ[1, 2, 2] $-r$	Γ[1, 3, 2] 0
Γ[1, 1, 3] 0	Γ[1, 2, 3] 0	Γ[1, 3, 3] $-r\,\text{Sin}[\theta]^2$
Γ[2, 1, 1] 0	Γ[2, 2, 1] $\frac{1}{r}$	Γ[2, 3, 1] 0
Γ[2, 1, 2] $\frac{1}{r}$	Γ[2, 2, 2] 0	Γ[2, 3, 2] 0
Γ[2, 1, 3] 0	Γ[2, 2, 3] 0	Γ[2, 3, 3] $-\text{Cos}[\theta]\,\text{Sin}[\theta]$
Γ[3, 1, 1] 0	Γ[3, 2, 1] 0	Γ[3, 3, 1] $\frac{1}{r}$
Γ[3, 1, 2] 0	Γ[3, 2, 2] 0	Γ[3, 3, 2] $\text{Cot}[\theta]$
Γ[3, 1, 3] $\frac{1}{r}$	Γ[3, 2, 3] $\text{Cot}[\theta]$	Γ[3, 3, 3] 0

```
listaffine :=
  Table[{Γ^ToString[i]_j,k, ToString["="], affine[[i, j, k]]},
   {i, 1, dim}, {j, 1, dim}, {k, 1, dim}]

listaffine // TableForm
```

$\Gamma^{1}{}_{1,1} = 0$	$\Gamma^{1}{}_{2,1} = 0$	$\Gamma^{1}{}_{3,1} = 0$
$\Gamma^{1}{}_{1,2} = 0$	$\Gamma^{1}{}_{2,2} = -r$	$\Gamma^{1}{}_{3,2} = 0$
$\Gamma^{1}{}_{1,3} = 0$	$\Gamma^{1}{}_{2,3} = 0$	$\Gamma^{1}{}_{3,3} = -r\,\text{Sin}[\theta]^2$
$\Gamma^{2}{}_{1,1} = 0$	$\Gamma^{2}{}_{2,1} = \frac{1}{r}$	$\Gamma^{2}{}_{3,1} = 0$
$\Gamma^{2}{}_{1,2} = \frac{1}{r}$	$\Gamma^{2}{}_{2,2} = 0$	$\Gamma^{2}{}_{3,2} = 0$
$\Gamma^{2}{}_{1,3} = 0$	$\Gamma^{2}{}_{2,3} = 0$	$\Gamma^{2}{}_{3,3} = -\text{Cos}[\theta]\,\text{Sin}[\theta]$
$\Gamma^{3}{}_{1,1} = 0$	$\Gamma^{3}{}_{2,1} = 0$	$\Gamma^{3}{}_{3,1} = \frac{1}{r}$
$\Gamma^{3}{}_{1,2} = 0$	$\Gamma^{3}{}_{2,2} = 0$	$\Gamma^{3}{}_{3,2} = \text{Cot}[\theta]$
$\Gamma^{3}{}_{1,3} = \frac{1}{r}$	$\Gamma^{3}{}_{2,3} = \text{Cot}[\theta]$	$\Gamma^{3}{}_{3,3} = 0$

Riemann Tensor

$$R^{i}{}_{jkl} = \Gamma^{i}{}_{jl,k} - \Gamma^{i}{}_{jk,l} + \Gamma^{i}{}_{sk}\Gamma^{s}{}_{jl} - \Gamma^{i}{}_{sl}\Gamma^{s}{}_{jk}$$

In[25]:=
```
Riemann := Riemann =
   Simplify[Table[D[affine[[i, j, l]], coord[[k]]] - D[affine[[i, j, k]], coord[[l]]] +
      Sum[affine[[i, s, k]] affine[[s, j, l]] - affine[[i, s, l]] affine[[s, j, k]],
       {s, 1, dim}],
     {i, 1, dim}, {j, 1, dim}, {k, 1, dim}, {l, 1, dim}]]
```

In[26]:=
```
Riemann[[1, 2, 1, 2]]
```

Out[26]= 0

In[37]:=
```
listR :=
  Table[{R^ToString[i]_j,k,l, Riemann[[i, j, k, l]]},
   {i, 1, dim}, {j, 1, dim}, {k, 1, dim}, {l, 1, dim}]
```

In[39]:=
```
TableForm[listR, TableSpacing → {3, 3}]
```

Out[39]//TableForm=

$R^1{}_{1,1,1}$	$R^1{}_{1,1,2}$	$R^1{}_{1,1,3}$	$R^1{}_{2,1,1}$	$R^1{}_{2,1,2}$	$R^1{}_{2,1,3}$	$R^1{}_{3,1,1}$	$R^1{}_{3,1,2}$	$R^1{}_{3,1,3}$
0	0	0	0	0	0	0	0	0
$R^1{}_{1,2,1}$	$R^1{}_{1,2,2}$	$R^1{}_{1,2,3}$	$R^1{}_{2,2,1}$	$R^1{}_{2,2,2}$	$R^1{}_{2,2,3}$	$R^1{}_{3,2,1}$	$R^1{}_{3,2,2}$	$R^1{}_{3,2,3}$
0	0	0	0	0	0	0	0	0
$R^1{}_{1,3,1}$	$R^1{}_{1,3,2}$	$R^1{}_{1,3,3}$	$R^1{}_{2,3,1}$	$R^1{}_{2,3,2}$	$R^1{}_{2,3,3}$	$R^1{}_{3,3,1}$	$R^1{}_{3,3,2}$	$R^1{}_{3,3,3}$
0	0	0	0	0	0	0	0	0
$R^2{}_{1,1,1}$	$R^2{}_{1,1,2}$	$R^2{}_{1,1,3}$	$R^2{}_{2,1,1}$	$R^2{}_{2,1,2}$	$R^2{}_{2,1,3}$	$R^2{}_{3,1,1}$	$R^2{}_{3,1,2}$	$R^2{}_{3,1,3}$
0	0	0	0	0	0	0	0	0
$R^2{}_{1,2,1}$	$R^2{}_{1,2,2}$	$R^2{}_{1,2,3}$	$R^2{}_{2,2,1}$	$R^2{}_{2,2,2}$	$R^2{}_{2,2,3}$	$R^2{}_{3,2,1}$	$R^2{}_{3,2,2}$	$R^2{}_{3,2,3}$
0	0	0	0	0	0	0	0	0
$R^2{}_{1,3,1}$	$R^2{}_{1,3,2}$	$R^2{}_{1,3,3}$	$R^2{}_{2,3,1}$	$R^2{}_{2,3,2}$	$R^2{}_{2,3,3}$	$R^2{}_{3,3,1}$	$R^2{}_{3,3,2}$	$R^2{}_{3,3,3}$
0	0	0	0	0	0	0	0	0
$R^3{}_{1,1,1}$	$R^3{}_{1,1,2}$	$R^3{}_{1,1,3}$	$R^3{}_{2,1,1}$	$R^3{}_{2,1,2}$	$R^3{}_{2,1,3}$	$R^3{}_{3,1,1}$	$R^3{}_{3,1,2}$	$R^3{}_{3,1,3}$
0	0	0	0	0	0	0	0	0
$R^3{}_{1,2,1}$	$R^3{}_{1,2,2}$	$R^3{}_{1,2,3}$	$R^3{}_{2,2,1}$	$R^3{}_{2,2,2}$	$R^3{}_{2,2,3}$	$R^3{}_{3,2,1}$	$R^3{}_{3,2,2}$	$R^3{}_{3,2,3}$
0	0	0	0	0	0	0	0	0
$R^3{}_{1,3,1}$	$R^3{}_{1,3,2}$	$R^3{}_{1,3,3}$	$R^3{}_{2,3,1}$	$R^3{}_{2,3,2}$	$R^3{}_{2,3,3}$	$R^3{}_{3,3,1}$	$R^3{}_{3,3,2}$	$R^3{}_{3,3,3}$
0	0	0	0	0	0	0	0	0

In[40]:=
```
Ricci := Table[Sum[Riemann[[u, a, u, b]], {u, 1, dim}], {a, 1, dim}, {b, 1, dim}]
```

In[42]:=
```
RicciR := Sum[Ricci[[a, b]] inversemetric[[a, b]], {a, 1, dim}, {b, 1, dim}]
```

Now put all these codes together.

```
dim = 3;
coord = {r, θ, ϕ};
metric = {{1, 0, 0}, {0, r^2, 0}, {0, 0, r^2 Sin[θ]^2}};
inversemetric = Simplify[Inverse[metric]];

affine := affine = Simplify[Table[(1/2) * Sum[(inversemetric[[i, s]]) *
          (D[metric[[s, j]], coord[[k]] ] +
            D[metric[[s, k]], coord[[j]] ] - D[metric[[j, k]], coord[[s]] ]),
         {s, 1, dim}],
       {i, 1, dim}, {j, 1, dim}, {k, 1, dim}] ];

Riemann := Riemann =
   Simplify[Table[D[affine[[i, j, l]], coord[[k]]] - D[affine[[i, j, k]], coord[[l]]] +
      Sum[affine[[i, s, k]] affine[[s, j, l]] - affine[[i, s, l]] affine[[s, j, k]],
        {s, 1, dim}],
      {i, 1, dim}, {j, 1, dim}, {k, 1, dim}, {l, 1, dim}]];

Ricci := Table[Sum[Riemann[[u, a, u, b]], {u, 1, dim}], {a, 1, dim}, {b, 1, dim}]

RicciR := Sum[Ricci[[a, b]] inversemetric[[a, b]], {a, 1, dim}, {b, 1, dim}]
```

Now solve below exercise.

Exercise.

Show that the 2-dimensional space with metric

$ds^2 = y\, dx^2 + x\, dy^2$

is curved, and that the curvature scalar is

$R = \frac{1}{2xy}\left(\frac{1}{x} + \frac{1}{y}\right)$

In[62]:=
```
dim = 2;
coord = {x, y};
metric = {{y, 0}, {0, x}};
inversemetric = Simplify[Inverse[metric]];
```

In[66]:=
```
affine := affine = Simplify[Table[(1 / 2) * Sum[(inversemetric[[i, s]]) *
        (D[metric[[s, j]], coord[[k]] ] +
          D[metric[[s, k]], coord[[j]] ] - D[metric[[j, k]], coord[[s]] ]),
       {s, 1, dim}],
     {i, 1, dim}, {j, 1, dim}, {k, 1, dim}] ];
```

In[67]:=
```
affine // TableForm
```

Out[67]//TableForm=

0 $\frac{1}{2y}$	$\frac{1}{2y}$ $-\frac{1}{2y}$
$-\frac{1}{2x}$ $\frac{1}{2x}$	$\frac{1}{2x}$ 0

In[68]:=
```
Riemann := Riemann =
   Simplify[Table[D[affine[[i, j, l]], coord[[k]]] - D[affine[[i, j, k]], coord[[l]]] +
      Sum[affine[[i, s, k]] affine[[s, j, l]] - affine[[i, s, l]] affine[[s, j, k]],
       {s, 1, dim}],
     {i, 1, dim}, {j, 1, dim}, {k, 1, dim}, {l, 1, dim}]];
```

In[69]:=
```
listR :=
  Table[{R^ToString[i]_j,k,l, Riemann[[i, j, k, l]]},
   {i, 1, dim}, {j, 1, dim}, {k, 1, dim}, {l, 1, dim}]
```

In[70]:=
```
listR // TableForm
```

Out[70]//TableForm=

${R^1}_{1,1,1}$ 0	${R^1}_{1,1,2}$ 0	${R^1}_{2,1,1}$ 0	${R^1}_{2,1,2}$ $\frac{x+y}{4\,x\,y^2}$
${R^1}_{1,2,1}$ 0	${R^1}_{1,2,2}$ 0	${R^1}_{2,2,1}$ $-\frac{x+y}{4\,x\,y^2}$	${R^1}_{2,2,2}$ 0
${R^2}_{1,1,1}$ 0	${R^2}_{1,1,2}$ $-\frac{x+y}{4\,x^2\,y}$	${R^2}_{2,1,1}$ 0	${R^2}_{2,1,2}$ 0
${R^2}_{1,2,1}$ $\frac{x+y}{4\,x^2\,y}$	${R^2}_{1,2,2}$ 0	${R^2}_{2,2,1}$ 0	${R^2}_{2,2,2}$ 0

In[71]:=
```
Ricci := Table[Sum[Riemann[[u, a, u, b]], {u, 1, dim}], {a, 1, dim}, {b, 1, dim}]
```

In[72]:=
```
Ricci // MatrixForm
```

Out[72]//MatrixForm=

$$\begin{pmatrix} \frac{x+y}{4 x^2 y} & 0 \\ 0 & \frac{x+y}{4 x y^2} \end{pmatrix}$$

In[73]:=
```
RicciR := Simplify[Sum[Ricci[[a, b]] inversemetric[[a, b]], {a, 1, dim}, {b, 1, dim}]]
```

In[74]:=
```
RicciR
```

Out[74]= $\frac{x+y}{2 x^2 y^2}$

In[75]:= Plot3D$\left[\frac{x+y}{2 x^2 y^2}, \{x, -8, 8\}, \{y, -8, 8\}\right]$

Out[75]=



We can guess that this space is approximately flat except for the neighborhoods of two axes.

Exercise.

A Riemannian space is said to be of constant curvature if the metric and Riemann tensors are such that

$R_{hijk} = K(g_{hj} g_{ik} - g_{hk} g_{ij})$

where K is a numerical constant. If the metric of such a space is

$ds^2 = dr^2 + f^2(r)\left(d\theta^2 + \sin^2\theta\, d\phi^2\right)$,

with $x^1 = r,\ x^2 = \theta,\ x^3 = \phi$ (holonomic basis), show that the three independent non-zero components of R_{hijk} are $R_{1212},\ R_{1313},\ R_{2323}$.

Calculate R_{1212} and show that K may be 1, 0, or -1 according as $f = \sin r,\ r$ or $\sinh r$ (up to multiplicative constant), provided that $f(0) = 0$

The total number of independent components of Riemann tensor

$\frac{n^2(n^2-1)}{12} = 6$ for $n = 3$, $R_{1212},\ R_{1213},\ R_{1223},\ R_{1313},\ R_{1323},\ R_{2323}$, $R_{abcd} = g_{ak} R^k{}_{bcd}$

In[76]:=
```
dim = 3;
coord = {r, θ, ϕ};
metric = {{1, 0, 0}, {0, (f[r])^2, 0}, {0, 0, (f[r] Sin[θ])^2}};
inversemetric = Simplify[Inverse[metric]];
```

In[81]:=
```
affine := affine = Simplify[Table[(1/2) * Sum[(inversemetric[[i, s]]) *
          (D[metric[[s, j]], coord[[k]] ] +
            D[metric[[s, k]], coord[[j]] ] - D[metric[[j, k]], coord[[s]] ]),
         {s, 1, dim}],
       {i, 1, dim}, {j, 1, dim}, {k, 1, dim}] ];
```

In[82]:=
```
TableForm[affine]
```

Out[82]//TableForm=

0 0 0	0 $-f[r]\, f'[r]$ 0	0 0 $-f[r]\, \mathrm{Sin}[\theta]^2 f'[r]$
0 $\frac{f'[r]}{f[r]}$ 0	$\frac{f'[r]}{f[r]}$ 0 0	0 0 $-\mathrm{Cos}[\theta]\, \mathrm{Sin}[\theta]$
0 0 $\frac{f'[r]}{f[r]}$	0 0 $\mathrm{Cot}[\theta]$	$\frac{f'[r]}{f[r]}$ $\mathrm{Cot}[\theta]$ 0

In[92]:=
```
Riemann := Riemann =
   Simplify[Table[D[affine[[i, j, l]], coord[[k]]] - D[affine[[i, j, k]], coord[[l]]] +
      Sum[affine[[i, s, k]] affine[[s, j, l]] - affine[[i, s, l]] affine[[s, j, k]],
       {s, 1, dim}],
     {i, 1, dim}, {j, 1, dim}, {k, 1, dim}, {l, 1, dim}]];
R = Table[Sum[metric[[a, k]] * Riemann[[k, b, c, d]], {k, 1, dim}],
   {a, 1, dim}, {b, 1, dim}, {c, 1, dim}, {d, 1, dim}];
```

In[94]:=
```
R[[1, 2, 1, 2]]
```

Out[94]= $-f[r]\, f''[r]$

$(g_{hj}\, g_{ik} - g_{hk}\, g_{ij})$, hijk – 1212

In[95]:=
```
metric[[1, 1]] metric[[2, 2]] - metric[[1, 2]] metric[[2, 1]]
```

Out[95]= $f[r]^2$

Now the problem reduced to find K such that

$-f(r)\, f''(r) = K\, f(r)^2$

- $f = \sin r \to K = 1$
- $f = r \to K = 0$
- $f = \sinh r \to K = -1$

Exercise.

A Riemann 4-space metric

$$ds^2 = e^{2\sigma}\left[(dx^1)^2 + (dx^2)^2 + (dx^3)^2 + (dx^4)^2\right]$$

where $\sigma = \sigma(x^1, x^2, x^3, x^4)$. If t^μ is the unit tangent to a geodesic prove that, along the geodesic,

$\frac{dt^\mu}{\delta\sigma} + 2\,(\sigma_\nu\, t^\nu)\, t^\mu = \sigma^\mu$, where $\sigma_\mu = \frac{\partial\sigma}{\partial x^\mu}$